Высшие симметрии

К лекции 5 (2024)

1 Определение D_x, D_x^{-1} и D_t, генерация однородных многочленов

Следующие функции уже использовались на лекции 2 (про законы сохранения). В них адаптируются встроенные функции D, Derivative и Integrate.

Пусть u[n] обозначает $\partial_x^n(u(x))$. Производная по x и интеграл:

In[19]:=

```
n2x = u[n_] ⧴ Derivative[n][u][x];
x2n = {u[x] → u[0], Derivative[n_][u][x] ⧴ u[n]};

dx[expr_] := D[expr /. n2x, x] /. x2n
dx[expr_, k_] := D[expr /. n2x, {x, k}] /. x2n

int[expr_] := Block[{i = Integrate[expr /. n2x, x]},
  If[FreeQ[i, Integrate], i /. x2n, i]]
```

Производная по t в силу уравнения $u_t = f$:

In[24]:=

```
n2t = u[n_] ⧴ Derivative[n, 0][u][x, t];
t2n = {u[x, t] → u[0], Derivative[n_, 0][u][x, t] ⧴ u[n]};

dt[expr_, f_] := D[expr /. n2t, t] /. u^(n_,1)[x, t] ⧴ D[f /. n2t, {x, n}] /. t2n

comm[f_, g_] := Expand[dt[g, f] - dt[f, g]]
```

vars[f, u] список переменных u_n в выражении

mon[d, r, k] список мономов степени d от u_k, u_{k+1}, ... с общей суммой индексов равной r

hom[m, n, M] список всех мономов заданного веса M относительно веса $u_j \sim m + j\,n$

rhom[m, n, M] то же самое, но без мономов, линейных по старшей переменной (следовательно, все мономы в этом списке неэквивалентны по модулю Im D_x)

In[28]:=

```
vars[f_, u__] := Union[Flatten[Cases[f, Blank[#], {0, Infinity}] & /@ {u}]]

mon[d_, r_, k_] := If[d == 1, If[r ≥ k, {u[r]}, {}],
  Flatten[Table[u[s] × mon[d - 1, r - s, s], {s, k, r}]]]

hom[m_, n_, M_] := Flatten[Table[
   If[IntegerQ[(M - m d) / n], mon[d, (M - m d) / n, 0], {}],
   {d, 1, Floor[M / m]}]]

rhom[m_, n_, M_] := Select[hom[m, n, M], Exponent[#, Last[vars[#, u]]] > 1 || # == u[0] &]
```

Тестовый пример: совместность КдФ с уравнением 5-го порядка:

In[32]:=
```
ut3 = u[3] - 6 u[0] × u[1]

ut5 = dx[u[4] - 10 u[0] × u[2] - 5 u[1]^2 + 10 u[0]^3]

dt[ut3, ut5]
dt[ut5, ut3]

Expand[% - %%]
```

Out[32]=
```
-6 u[0] × u[1] + u[3]
```

Out[33]=
```
30 u[0]^2 u[1] - 20 u[1] × u[2] - 10 u[0] × u[3] + u[5]
```

Out[34]=
```
90 u[2] (2 u[1]^2 + 2 u[0] × u[2]) + 180 u[0] × u[1] × u[3] - 70 u[3]^2 +
 30 u[1] (6 u[1] × u[2] + 2 u[0] × u[3]) + 30 u[0]^2 u[4] - 110 u[2] × u[4] - 50 u[1] × u[5] -
 6 u[1] (30 u[0]^2 u[1] - 20 u[1] × u[2] - 10 u[0] × u[3] + u[5]) - 10 u[0] × u[6] -
 6 u[0] (60 u[0] u[1]^2 + 30 u[0]^2 u[2] - 20 u[2]^2 - 30 u[1] × u[3] - 10 u[0] × u[4] + u[6]) + u[8]
```

Out[35]=
```
-60 u[3]^2 + 60 u[0] × u[1] (-6 u[0] × u[1] + u[3]) - 10 u[3] (-6 u[0] × u[1] + u[3]) - 90 u[2] × u[4] +
 30 u[0]^2 (-6 u[1]^2 - 6 u[0] × u[2] + u[4]) - 20 u[2] (-6 u[1]^2 - 6 u[0] × u[2] + u[4]) -
 36 u[1] × u[5] - 20 u[1] (-18 u[1] × u[2] - 6 u[0] × u[3] + u[5]) -
 6 u[0] × u[6] - 10 u[0] (-18 u[2]^2 - 24 u[1] × u[3] - 6 u[0] × u[4] + u[6]) + u[8]
```

Out[36]=
```
0
```

2 Иерархия Бюргерса

Совместность линейных уравнений очевидна

In[37]:=
```
dt[u[m], u[n]]
```

Out[37]=
```
u[m + n]
```

Понятно (?), что при точечных заменах и подстановках совместность не портится. Сделаем замену к уравнению Бюргерса:

$$u = \exp(v), \quad u^B = v_x = u_x / u$$

In[38]:=
```
uBt[n_] := Expand[dx[Exp[-u[0]] dx[Exp[u[0]], n]]] /. u[k_] :→ u[k - 1]

uBt[1]
uBt[2]
uBt[3]
uBt[4]
```

Out[39]=
```
u[1]
```

Out[40]=
```
2 u[0] × u[1] + u[2]
```

Out[41]=
```
3 u[0]^2 u[1] + 3 u[1]^2 + 3 u[0] × u[2] + u[3]
```

Out[42]=
```
4 u[0]^3 u[1] + 12 u[0] u[1]^2 + 6 u[0]^2 u[2] + 10 u[1] × u[2] + 4 u[0] × u[3] + u[4]
```

Действительно, коммутаторы равны 0:

In[43]:=
```
comm[uBt[2], uBt[3]]
comm[uBt[2], uBt[4]]
comm[uBt[3], uBt[4]]
```

Out[43]=
```
0
```

Out[44]=
```
0
```

Out[45]=
```
0
```

Оператор рекурсии $R = D + u + u_1 D^{-1}$

In[46]:=
```
R[f_] := Expand[dx[f] + u[0] f + u[1] × int[f]]

Ut[1] = uBt[1]
Ut[2] = R[%]
Ut[3] = R[%]
Ut[4] = R[%]
Ut[5] = R[%]
```

Out[47]=
```
u[1]
```

Out[48]=
```
2 u[0] × u[1] + u[2]
```

Out[49]=
```
3 u[0]^2 u[1] + 3 u[1]^2 + 3 u[0] × u[2] + u[3]
```

Out[50]=
```
4 u[0]^3 u[1] + 12 u[0] u[1]^2 + 6 u[0]^2 u[2] + 10 u[1] × u[2] + 4 u[0] × u[3] + u[4]
```

Out[51]=
```
5 u[0]^4 u[1] + 30 u[0]^2 u[1]^2 + 15 u[1]^3 + 10 u[0]^3 u[2] +
 50 u[0] × u[1] × u[2] + 10 u[2]^2 + 10 u[0]^2 u[3] + 15 u[1] × u[3] + 5 u[0] × u[4] + u[5]
```

3 Вычисление симметрий КдФ методом неопределенных

коэффициентов

Вычисление симметрий 5-го и 7-го порядков

Составим общий однородный многочлен веса 7; коэффициент при u_5 положим равным 1

In[75]:=
```
ut3 = u[3] - 6 u[0] × u[1];

hom[2, 1, 7]
a /@ Range[Length[%]]

uT = %.%% /. a[1] → 1
```

Out[76]=
```
{u[5], u[0] × u[3], u[1] × u[2], u[0]^2 u[1]}
```

Out[77]=
```
{a[1], a[2], a[3], a[4]}
```

Out[78]=
```
a[4] u[0]^2 u[1] + a[3] × u[1] × u[2] + a[2] × u[0] × u[3] + u[5]
```

Составим уравнение $[u_{t_3}, u_T] = 0$, соберем коэффициенты, решим систему, по решению определим u_{t_5}

In[79]:=
```
eq = dt[uT, ut3] - dt[ut3, uT]
vars[eq, u]
Union[Flatten[CoefficientList[eq, %]]]
sol = Solve[% == 0]

ut5 = uT /. sol[[1]]
```

Out[79]=
```
-3 a[4] × u[2] (2 u[1]^2 + 2 u[0] × u[2]) - 6 a[4] × u[0] × u[1] × u[3] - 60 u[3]^2 - a[2] u[3]^2 -
 3 a[3] u[3]^2 + 2 a[4] × u[0] × u[1] (-6 u[0] × u[1] + u[3]) + a[2] × u[3] (-6 u[0] × u[1] + u[3]) -
 a[4] × u[1] (6 u[1] × u[2] + 2 u[0] × u[3]) - a[4] u[0]^2 u[4] - 90 u[2] × u[4] -
 3 a[2] × u[2] × u[4] - 4 a[3] × u[2] × u[4] + a[4] u[0]^2 (-6 u[1]^2 - 6 u[0] × u[2] + u[4]) +
 a[3] × u[2] (-6 u[1]^2 - 6 u[0] × u[2] + u[4]) - 36 u[1] × u[5] - 3 a[2] × u[1] × u[5] -
 a[3] × u[1] × u[5] + a[3] × u[1] (-18 u[1] × u[2] - 6 u[0] × u[3] + u[5]) +
 6 u[1] (a[4] u[0]^2 u[1] + a[3] × u[1] × u[2] + a[2] × u[0] × u[3] + u[5]) - 6 u[0] × u[6] -
 a[2] × u[0] × u[6] + a[2] × u[0] (-18 u[2]^2 - 24 u[1] × u[3] - 6 u[0] × u[4] + u[6]) +
 6 u[0] (2 a[4] × u[0] u[1]^2 + a[4] u[0]^2 u[2] + a[3] u[2]^2 +
    a[2] × u[1] × u[3] + a[3] × u[1] × u[3] + a[2] × u[0] × u[4] + u[6])
```

Out[80]=
```
{u[0], u[1], u[2], u[3], u[4], u[5], u[6]}
```

Out[81]=
```
{0, -30 - 3 a[2], -60 - 3 a[3], -90 - 3 a[2] - 3 a[3], -18 a[3] - 12 a[4], -18 a[2] - 6 a[4]}
```

Out[82]=
```
{{a[2] → -10, a[3] → -20, a[4] → 30}}
```

Out[83]=
```
30 u[0]^2 u[1] - 20 u[1] × u[2] - 10 u[0] × u[3] + u[5]
```

Проверка

In[84]:=
```
comm[ut5, ut3]
```

Out[84]=

```
0
```

То же самое проделаем с однородным многочленом веса 9.

In[85]:=
```
hom[2, 1, 9]
a /@ Range[Length[%]]

uT = %.%% /. a[1] → 1
```

Out[85]=
```
{u[7], u[0]×u[5], u[1]×u[4], u[2]×u[3], u[0]^2 u[3], u[0]×u[1]×u[2], u[1]^3, u[0]^3 u[1]}
```

Out[86]=
```
{a[1], a[2], a[3], a[4], a[5], a[6], a[7], a[8]}
```

Out[87]=
```
a[8] u[0]^3 u[1] + a[7] u[1]^3 + a[6]×u[0]×u[1]×u[2] +
 a[5] u[0]^2 u[3] + a[4]×u[2]×u[3] + a[3]×u[1]×u[4] + a[2]×u[0]×u[5] + u[7]
```

In[88]:=
```
eq = comm[ut3, uT]
vars[eq, u]
Union[Flatten[CoefficientList[eq, %]]]
sol = Solve[% == 0]

ut7 = uT /. sol[[1]]
```

Out[88]=
```
-12 a[7] u[1]^4 - 6 a[8] u[1]^4 - 18 a[6]×u[0] u[1]^2 u[2] - 36 a[8]×u[0] u[1]^2 u[2] -
 18 a[5] u[0]^2 u[2]^2 - 9 a[8] u[0]^2 u[2]^2 - 18 a[4] u[2]^3 - 3 a[6] u[2]^3 - 6 a[7] u[2]^3 -
 18 a[5] u[0]^2 u[1]×u[3] - 9 a[8] u[0]^2 u[1]×u[3] - 60 a[3]×u[1]×u[2]×u[3] -
 36 a[4]×u[1]×u[2]×u[3] - 6 a[5]×u[1]×u[2]×u[3] - 12 a[6]×u[1]×u[2]×u[3] -
 18 a[7]×u[1]×u[2]×u[3] - 60 a[2]×u[0] u[3]^2 - 3 a[6]×u[0] u[3]^2 - 30 a[3] u[1]^2 u[4] -
 6 a[5] u[1]^2 u[4] - 3 a[6] u[1]^2 u[4] - 90 a[2]×u[0]×u[2]×u[4] - 6 a[5]×u[0]×u[2]×u[4] -
 3 a[6]×u[0]×u[2]×u[4] - 210 u[4]^2 - 3 a[4] u[4]^2 - 30 a[2]×u[0]×u[1]×u[5] -
 6 a[5]×u[0]×u[1]×u[5] - 336 u[3]×u[5] - 3 a[3]×u[3]×u[5] - 3 a[4]×u[3]×u[5] -
 168 u[2]×u[6] - 3 a[2]×u[2]×u[6] - 3 a[3]×u[2]×u[6] - 42 u[1]×u[7] - 3 a[2]×u[1]×u[7]
```

Out[89]=
```
{u[0], u[1], u[2], u[3], u[4], u[5], u[6], u[7]}
```

Out[90]=
```
{0, -42 - 3 a[2], -168 - 3 a[2] - 3 a[3], -210 - 3 a[4],
 -336 - 3 a[3] - 3 a[4], -30 a[2] - 6 a[5], -60 a[2] - 3 a[6], -90 a[2] - 6 a[5] - 3 a[6],
 -30 a[3] - 6 a[5] - 3 a[6], -60 a[3] - 36 a[4] - 6 a[5] - 12 a[6] - 18 a[7],
 -18 a[4] - 3 a[6] - 6 a[7], -18 a[6] - 36 a[8], -18 a[5] - 9 a[8], -12 a[7] - 6 a[8]}
```

Out[91]=
```
{{a[2] → -14, a[3] → -42, a[4] → -70, a[5] → 70, a[6] → 280, a[7] → 70, a[8] → -140}}
```

Out[92]=
```
-140 u[0]^3 u[1] + 70 u[1]^3 + 280 u[0]×u[1]×u[2] +
 70 u[0]^2 u[3] - 70 u[2]×u[3] - 42 u[1]×u[4] - 14 u[0]×u[5] + u[7]
```

Проверка

In[93]:=
```
comm[ut7, ut3]
comm[ut7, ut5]
```

Out[93]=
```
0
```

Out[94]=
```
0
```

Вычисление в процедуре

Сделаем то же самое в виде процедуры для любого веса.

In[95]:=
```
ut3 = u[3] - 6 u[0] × u[1];

sym[m_] := Module[{uT, a, eq},
  uT = hom[2, 1, m + 2];
  uT = Table[a[i], {i, 1, Length[uT]}].uT /. a[1] → 1;
  eq = comm[uT, ut3];
  eq = Union[Flatten[{CoefficientList[eq, vars[eq, u]]}]];
  eq = Solve[eq == 0];
  If[Length[eq] == 0, 0, uT /. eq[[1]]]
 ]
```

Симметрий чётного порядка нет:

In[97]:=
```
sym[2]
sym[4]
sym[6]
```

Out[97]=
```
0
```

Out[98]=
```
0
```

Out[99]=
```
0
```

Симметрии нечётного порядка:

In[100]:=

```
uKt[1] = sym[1]
uKt[3] = sym[3]
uKt[5] = sym[5]
uKt[7] = sym[7]
uKt[9] = sym[9]
uKt[11] = sym[11]
```

Out[100]=

u[1]

Out[101]=

-6 u[0] × u[1] + u[3]

Out[102]=

30 u[0]^2 u[1] - 20 u[1] × u[2] - 10 u[0] × u[3] + u[5]

Out[103]=

-140 u[0]^3 u[1] + 70 u[1]^3 + 280 u[0] × u[1] × u[2] +
70 u[0]^2 u[3] - 70 u[2] × u[3] - 42 u[1] × u[4] - 14 u[0] × u[5] + u[7]

Out[104]=

630 u[0]^4 u[1] - 1260 u[0] u[1]^3 - 2520 u[0]^2 u[1] × u[2] + 1302 u[1] u[2]^2 -
420 u[0]^3 u[3] + 966 u[1]^2 u[3] + 1260 u[0] × u[2] × u[3] + 756 u[0] × u[1] × u[4] -
252 u[3] × u[4] + 126 u[0]^2 u[5] - 168 u[2] × u[5] - 72 u[1] × u[6] - 18 u[0] × u[7] + u[9]

Out[105]=

-2772 u[0]^5 u[1] + 13 860 u[0]^2 u[1]^3 + 18 480 u[0]^3 u[1] × u[2] -
14 784 u[1]^3 u[2] - 28 644 u[0] × u[1] u[2]^2 + 2310 u[0]^4 u[3] - 21 252 u[0] u[1]^2 u[3] -
13 860 u[0]^2 u[2] × u[3] + 9702 u[2]^2 u[3] + 7194 u[1] u[3]^2 - 8316 u[0]^2 u[1] × u[4] +
11 484 u[1] × u[2] × u[4] + 5544 u[0] × u[3] × u[4] - 924 u[0]^3 u[5] + 2838 u[1]^2 u[5] +
3696 u[0] × u[2] × u[5] - 924 u[4] × u[5] + 1584 u[0] × u[1] × u[6] - 660 u[3] × u[6] +
198 u[0]^2 u[7] - 330 u[2] × u[7] - 110 u[1] × u[8] - 22 u[0] × u[9] + u[11]

Коммутативность симметрий:

In[106]:=

```
comm[uKt[5], uKt[9]]
comm[uKt[7], uKt[9]]
```

Out[106]=

0

Out[107]=

0

Симметрия 15-го порядка вычисляется довольно долго:

In[108]:=

```
sym[15]
```

Out[108]=

$$-51\,480\,u[0]^7\,u[1] + 900\,900\,u[0]^4\,u[1]^3 - 900\,900\,u[0]\,u[1]^5 + 720\,720\,u[0]^5\,u[1]\times u[2] -$$
$$5\,765\,760\,u[0]^2\,u[1]^3\,u[2] - 3\,723\,720\,u[0]^3\,u[1]\,u[2]^2 + 4\,178\,460\,u[1]^3\,u[2]^2 +$$
$$5\,319\,600\,u[0]\times u[1]\,u[2]^3 + 60\,060\,u[0]^6\,u[3] - 2\,762\,760\,u[0]^3\,u[1]^2\,u[3] + 1\,540\,110\,u[1]^4\,u[3] -$$
$$900\,900\,u[0]^4\,u[2]\times u[3] + 11\,788\,920\,u[0]\,u[1]^2\,u[2]\times u[3] + 3\,783\,780\,u[0]^2\,u[2]^2\,u[3] -$$
$$2\,279\,420\,u[2]^3\,u[3] + 2\,805\,660\,u[0]^2\,u[1]\,u[3]^2 - 5\,045\,040\,u[1]\times u[2]\,u[3]^2 - 797\,940\,u[0]\,u[3]^3 -$$
$$540\,540\,u[0]^4\,u[1]\times u[4] + 2\,316\,600\,u[0]\,u[1]^3\,u[4] + 4\,478\,760\,u[0]^2\,u[1]\times u[2]\times u[4] -$$
$$4\,010\,292\,u[1]\,u[2]^2\,u[4] + 720\,720\,u[0]^3\,u[3]\times u[4] - 2\,961\,816\,u[1]^2\,u[3]\times u[4] -$$
$$3\,809\,520\,u[0]\times u[2]\times u[3]\times u[4] - 1\,123\,980\,u[0]\times u[1]\,u[4]^2 + 578\,006\,u[3]\,u[4]^2 -$$
$$36\,036\,u[0]^5\,u[5] + 1\,106\,820\,u[0]^2\,u[1]^2\,u[5] + 480\,480\,u[0]^3\,u[2]\times u[5] - 1\,959\,672\,u[1]^2\,u[2]\times u[5] -$$
$$1\,261\,260\,u[0]\,u[2]^2\,u[5] - 1\,870\,440\,u[0]\times u[1]\times u[3]\times u[5] + 480\,194\,u[3]^2\,u[5] -$$
$$360\,360\,u[0]^2\,u[4]\times u[5] + 763\,620\,u[2]\times u[4]\times u[5] + 187\,174\,u[1]\,u[5]^2 + 205\,920\,u[0]^3\,u[1]\times u[6] -$$
$$274\,560\,u[1]^3\,u[6] - 1\,063\,920\,u[0]\times u[1]\times u[2]\times u[6] - 257\,400\,u[0]^2\,u[3]\times u[6] +$$
$$543\,400\,u[2]\times u[3]\times u[6] + 320\,632\,u[1]\times u[4]\times u[6] + 102\,960\,u[0]\times u[5]\times u[6] +$$
$$12\,870\,u[0]^4\,u[7] - 197\,340\,u[0]\,u[1]^2\,u[7] - 128\,700\,u[0]^2\,u[2]\times u[7] + 134\,706\,u[2]^2\,u[7] +$$
$$199\,888\,u[1]\times u[3]\times u[7] + 77\,220\,u[0]\times u[4]\times u[7] - 12\,870\,u[6]\times u[7] - 42\,900\,u[0]^2\,u[1]\times u[8] +$$
$$88\,400\,u[1]\times u[2]\times u[8] + 42\,900\,u[0]\times u[3]\times u[8] - 10\,010\,u[5]\times u[8] - 2860\,u[0]^3\,u[9] +$$
$$13\,130\,u[1]^2\,u[9] + 17\,160\,u[0]\times u[2]\times u[9] - 6006\,u[4]\times u[9] + 4680\,u[0]\times u[1]\times u[10] -$$
$$2730\,u[3]\times u[10] + 390\,u[0]^2\,u[11] - 910\,u[2]\times u[11] - 210\,u[1]\times u[12] - 30\,u[0]\times u[13] + u[15]$$

4 Уравнения SK и KK

Попробуем поискать одновременно $u_t = f$ и $u_T = g$

u_t порядка 4, u_T порядка 6

In[109]:=

```
hom[2, 1, 6]
a /@ Range[Length[%]];
f = %.%% /. a[1] → 1

hom[2, 1, 8]
b /@ Range[Length[%]];
g = %.%% /. b[1] → 1
```

Out[109]=

$$\{u[4],\ u[0]\times u[2],\ u[1]^2,\ u[0]^3\}$$

Out[111]=

$$a[4]\,u[0]^3 + a[3]\,u[1]^2 + a[2]\times u[0]\times u[2] + u[4]$$

Out[112]=

$$\{u[6],\ u[0]\times u[4],\ u[1]\times u[3],\ u[2]^2,\ u[0]^2\,u[2],\ u[0]\,u[1]^2,\ u[0]^4\}$$

Out[114]=

$$b[7]\,u[0]^4 + b[6]\times u[0]\,u[1]^2 + b[5]\,u[0]^2\,u[2] +$$
$$b[4]\,u[2]^2 + b[3]\times u[1]\times u[3] + b[2]\times u[0]\times u[4] + u[6]$$

Система на коэффициенты получается нелинейная, но *Mathematica* с ней справляется. Решений нет.

In[115]:=

```
eq = comm[f, g]
vars[eq, u]
Union[Flatten[CoefficientList[eq, %]]]
sol = Solve[% == 0]
```

Out[115]=

a[4] × b[7] u[0]^6 + 6 a[4] × b[5] u[0]^3 u[1]^2 + 4 a[4] × b[6] u[0]^3 u[1]^2 - 12 a[2] × b[7] u[0]^3 u[1]^2 -
4 a[3] × b[7] u[0]^3 u[1]^2 + 6 a[4] × b[3] u[1]^4 - a[3] × b[6] u[1]^4 - 24 b[7] u[1]^4 +
2 a[4] × b[5] u[0]^4 u[2] - a[2] × b[7] u[0]^4 u[2] + 36 a[4] × b[2] × u[0] u[1]^2 u[2] +
18 a[4] × b[3] × u[0] u[1]^2 u[2] + 12 a[4] × b[4] × u[0] u[1]^2 u[2] - 2 a[2] × b[5] × u[0] u[1]^2 u[2] -
2 a[3] × b[5] × u[0] u[1]^2 u[2] - 3 a[2] × b[6] × u[0] u[1]^2 u[2] - 144 b[7] × u[0] u[1]^2 u[2] +
18 a[4] × b[2] u[0]^2 u[2]^2 + 3 a[4] × b[4] u[0]^2 u[2]^2 + 2 a[3] × b[5] u[0]^2 u[2]^2 -
2 a[2] × b[6] u[0]^2 u[2]^2 - 36 b[7] u[0]^2 u[2]^2 + 90 a[4] u[2]^3 + a[2] × b[4] u[2]^3 +
4 a[3] × b[4] u[2]^3 - 6 b[5] u[2]^3 - 12 b[6] u[2]^3 + 24 a[4] × b[2] u[0]^2 u[1] × u[3] +
3 a[4] × b[3] u[0]^2 u[1] × u[3] - 2 a[2] × b[5] u[0]^2 u[1] × u[3] - 48 b[7] u[0]^2 u[1] × u[3] +
360 a[4] × u[1] × u[2] × u[3] + 4 a[2] × b[3] × u[1] × u[2] × u[3] + 6 a[3] × b[3] × u[1] × u[2] × u[3] +
4 a[2] × b[4] × u[1] × u[2] × u[3] - 32 b[5] × u[1] × u[2] × u[3] - 44 b[6] × u[1] × u[2] × u[3] +
60 a[4] × u[0] u[3]^2 + 4 a[2] × b[2] × u[0] u[3]^2 + 6 a[3] × b[2] × u[0] u[3]^2 -
2 a[2] × b[4] × u[0] u[3]^2 - 8 b[5] × u[0] u[3]^2 - 6 b[6] × u[0] u[3]^2 + a[4] × b[2] u[0]^3 u[4] +
90 a[4] u[1]^2 u[4] - a[3] × b[2] u[1]^2 u[4] + 3 a[2] × b[3] u[1]^2 u[4] - 12 b[5] u[1]^2 u[4] -
8 b[6] u[1]^2 u[4] + 90 a[4] × u[0] × u[2] × u[4] + 6 a[2] × b[2] × u[0] × u[2] × u[4] +
8 a[3] × b[2] × u[0] × u[2] × u[4] - 2 a[2] × b[3] × u[0] × u[2] × u[4] - 12 b[5] × u[0] × u[2] × u[4] -
8 b[6] × u[0] × u[2] × u[4] + 15 a[2] u[4]^2 + 20 a[3] u[4]^2 - 4 b[3] u[4]^2 -
6 b[4] u[4]^2 + 36 a[4] × u[0] × u[1] × u[5] + 2 a[2] × b[2] × u[0] × u[1] × u[5] -
8 b[5] × u[0] × u[1] × u[5] + 26 a[2] × u[3] × u[5] + 30 a[3] × u[3] × u[5] - 4 b[2] × u[3] × u[5] -
6 b[3] × u[3] × u[5] - 8 b[4] × u[3] × u[5] + 15 a[2] × u[2] × u[6] + 12 a[3] × u[2] × u[6] -
6 b[2] × u[2] × u[6] - 4 b[3] × u[2] × u[6] + 6 a[2] × u[1] × u[7] - 4 b[2] × u[1] × u[7]

Out[116]=

{u[0], u[1], u[2], u[3], u[4], u[5], u[6], u[7]}

Out[117]=

{0, 6 a[2] - 4 b[2], a[4] × b[2],
15 a[2] + 12 a[3] - 6 b[2] - 4 b[3], 26 a[2] + 30 a[3] - 4 b[2] - 6 b[3] - 8 b[4],
15 a[2] + 20 a[3] - 4 b[3] - 6 b[4], 36 a[4] + 2 a[2] × b[2] - 8 b[5],
360 a[4] + 4 a[2] × b[3] + 6 a[3] × b[3] + 4 a[2] × b[4] - 32 b[5] - 44 b[6],
90 a[4] + a[2] × b[4] + 4 a[3] × b[4] - 6 b[5] - 12 b[6],
90 a[4] + 6 a[2] × b[2] + 8 a[3] × b[2] - 2 a[2] × b[3] - 12 b[5] - 8 b[6],
90 a[4] - a[3] × b[2] + 3 a[2] × b[3] - 12 b[5] - 8 b[6],
60 a[4] + 4 a[2] × b[2] + 6 a[3] × b[2] - 2 a[2] × b[4] - 8 b[5] - 6 b[6],
36 a[4] × b[2] + 18 a[4] × b[3] + 12 a[4] × b[4] - 2 a[2] × b[5] - 2 a[3] × b[5] - 3 a[2] × b[6] - 144 b[7],
24 a[4] × b[2] + 3 a[4] × b[3] - 2 a[2] × b[5] - 48 b[7],
18 a[4] × b[2] + 3 a[4] × b[4] + 2 a[3] × b[5] - 2 a[2] × b[6] - 36 b[7],
6 a[4] × b[3] - a[3] × b[6] - 24 b[7], a[4] × b[7], 2 a[4] × b[5] - a[2] × b[7],
6 a[4] × b[5] + 4 a[4] × b[6] - 12 a[2] × b[7] - 4 a[3] × b[7]}

Out[118]=

{{a[2] → 0, a[3] → 0, a[4] → 0, b[2] → 0, b[3] → 0, b[4] → 0, b[5] → 0, b[6] → 0, b[7] → 0}}

u_t порядка 5, u_T порядка 7

In[119]:=

```
hom[2, 1, 7]
a /@ Range[Length[%]];
f = %.%% /. a[1] → 1

hom[2, 1, 9]
b /@ Range[Length[%]];
g = %.%% /. b[1] → 1
```

Out[119]=

{u[5], u[0] × u[3], u[1] × u[2], u[0]^2 u[1]}

Out[121]=

a[4] u[0]^2 u[1] + a[3] × u[1] × u[2] + a[2] × u[0] × u[3] + u[5]

Out[122]=

{u[7], u[0] × u[5], u[1] × u[4], u[2] × u[3], u[0]^2 u[3], u[0] × u[1] × u[2], u[1]^3, u[0]^3 u[1]}

Out[124]=

b[8] u[0]^3 u[1] + b[7] u[1]^3 + b[6] × u[0] × u[1] × u[2] +
b[5] u[0]^2 u[3] + b[4] × u[2] × u[3] + b[3] × u[1] × u[4] + b[2] × u[0] × u[5] + u[7]

Находится 3 решения!

In[125]:=

```
eq = comm[f, g];
vars[eq, u];
Union[Flatten[CoefficientList[eq, %]]]
sol = Solve[% == 0]
```

Out[127]=

$\{0, 7 a[2] - 5 b[2], 21 a[2] + 7 a[3] - 10 b[2] - 5 b[3],$
$42 a[2] + 56 a[3] - 5 b[2] - 10 b[3] - 15 b[4], 21 a[2] + 35 a[3] - 5 b[3] - 10 b[4],$
$35 a[2] + 28 a[3] - 10 b[2] - 10 b[3] - 5 b[4], 14 a[4] + 2 a[2] \times b[2] - 10 b[5],$
$112 a[4] + 10 a[2] \times b[2] + 15 a[3] \times b[2] - 3 a[2] \times b[3] - 3 a[2] \times b[4] - 20 b[5] - 15 b[6],$
$70 a[4] + 5 a[2] \times b[2] + 10 a[3] \times b[2] - 3 a[2] \times b[4] - 10 b[5] - 10 b[6],$
$56 a[4] + 7 a[2] \times b[2] + 5 a[3] \times b[2] - 3 a[2] \times b[3] - 20 b[5] - 5 b[6],$
$56 a[4] - 2 a[3] \times b[2] + 4 a[2] \times b[3] - 20 b[5] - 5 b[6],$
$560 a[4] + a[2] \times b[4] + 5 a[3] \times b[4] - 20 b[5] - 60 b[6] - 90 b[7],$
$560 a[4] + 5 a[2] \times b[3] + 10 a[3] \times b[3] + 2 a[2] \times b[4] - 2 a[3] \times b[4] - 50 b[5] - 65 b[6] - 60 b[7],$
$420 a[4] + 3 a[2] \times b[4] + 3 a[3] \times b[4] - 30 b[5] - 45 b[6] - 60 b[7],$
$336 a[4] - a[3] \times b[2] + 6 a[2] \times b[3] + 2 a[3] \times b[3] + 3 a[2] \times b[4] - 60 b[5] - 35 b[6] - 30 b[7],$
$120 a[4] \times b[2] + 20 a[4] \times b[3] + 12 a[4] \times b[4] - 6 a[2] \times b[5] - 2 a[3] \times b[5] - 11 a[2] \times b[6] -$
$18 a[2] \times b[7] - 360 b[8], 30 a[4] \times b[3] + 12 a[4] \times b[4] - 3 a[3] \times b[6] - 6 a[3] \times b[7] - 270 b[8],$
$20 a[4] \times b[3] + 2 a[4] \times b[4] - 2 a[3] \times b[5] - 2 a[3] \times b[6] + 2 a[2] \times b[7] - 120 b[8],$
$30 a[4] \times b[2] + 10 a[4] \times b[3] - 6 a[2] \times b[5] - 4 a[3] \times b[5] - a[2] \times b[6] - 90 b[8],$
$30 a[4] \times b[2] + 6 a[4] \times b[4] - 3 a[2] \times b[6] - 6 a[2] \times b[7] - 90 b[8],$
$30 a[4] \times b[2] - 3 a[2] \times b[5] + 3 a[3] \times b[5] - 3 a[2] \times b[6] - 45 b[8],$
$20 a[4] \times b[2] + 3 a[3] \times b[5] - 3 a[2] \times b[6] - 30 b[8], 10 a[4] \times b[2] - 3 a[2] \times b[5] - 15 b[8],$
$6 a[4] \times b[5] - 9 a[2] \times b[8], 12 a[4] \times b[5] + 6 a[4] \times b[6] - 36 a[2] \times b[8] - 9 a[3] \times b[8],$
$2 a[4] \times b[6] + 4 a[4] \times b[7] - 6 a[2] \times b[8] - 6 a[3] \times b[8]\}$

Out[128]=

$$\Big\{\Big\{a[3] \to a[2],\ a[4] \to \frac{a[2]^2}{5},\ b[2] \to \frac{7 a[2]}{5},\ b[3] \to \frac{14 a[2]}{5},$$
$$b[4] \to \frac{21 a[2]}{5},\ b[5] \to \frac{14 a[2]^2}{25},\ b[6] \to \frac{42 a[2]^2}{25},\ b[7] \to \frac{7 a[2]^2}{25},\ b[8] \to \frac{28 a[2]^3}{375}\Big\},$$
$$\Big\{a[3] \to \frac{5 a[2]}{2},\ a[4] \to \frac{a[2]^2}{5},\ b[2] \to \frac{7 a[2]}{5},\ b[3] \to \frac{49 a[2]}{10},\ b[4] \to \frac{42 a[2]}{5},$$
$$b[5] \to \frac{14 a[2]^2}{25},\ b[6] \to \frac{63 a[2]^2}{25},\ b[7] \to \frac{7 a[2]^2}{10},\ b[8] \to \frac{28 a[2]^3}{375}\Big\},$$
$$\Big\{a[3] \to 2 a[2],\ a[4] \to \frac{3 a[2]^2}{10},\ b[2] \to \frac{7 a[2]}{5},\ b[3] \to \frac{21 a[2]}{5},\ b[4] \to 7 a[2],$$
$$b[5] \to \frac{7 a[2]^2}{10},\ b[6] \to \frac{14 a[2]^2}{5},\ b[7] \to \frac{7 a[2]^2}{10},\ b[8] \to \frac{7 a[2]^3}{50}\Big\}\Big\}$$

Одно из них совпадает с симметрией КдФ, при подходящем выборе свободного коэффициента (который отвечает за растяжение)

In[129]:=

```
uKt[5]
f /. sol[[3]] /. a[2] → -10
```

Out[129]=

$30 u[0]^2 u[1] - 20 u[1] \times u[2] - 10 u[0] \times u[3] + u[5]$

Out[130]=

$30 u[0]^2 u[1] - 20 u[1] \times u[2] - 10 u[0] \times u[3] + u[5]$

Два других – новые уравнения, не связанные с КдФ. Уравнение Савады-Котеры

In[131]:=

```
uSKt[5] = f /. sol⟦1⟧ /. a[2] → 5
uSKt[7] = g /. sol⟦1⟧ /. a[2] → 5
comm[%, %%]
```

Out[131]=

$5\,u[0]^2\,u[1] + 5\,u[1]\times u[2] + 5\,u[0]\times u[3] + u[5]$

Out[132]=

$\frac{28}{3}\,u[0]^3\,u[1] + 7\,u[1]^3 + 42\,u[0]\times u[1]\times u[2] + 14\,u[0]^2\,u[3] + 21\,u[2]\times u[3] + 14\,u[1]\times u[4] + 7\,u[0]\times u[5] + u[7]$

Out[133]=

0

и Каупа-Купершмидта

In[134]:=

```
uKKt[5] = f /. sol⟦2⟧ /. a[2] → 5
uKKt[7] = g /. sol⟦2⟧ /. a[2] → 5
comm[%, %%]
```

Out[134]=

$5\,u[0]^2\,u[1] + \frac{25}{2}\,u[1]\times u[2] + 5\,u[0]\times u[3] + u[5]$

Out[135]=

$\frac{28}{3}\,u[0]^3\,u[1] + \frac{35\,u[1]^3}{2} + 63\,u[0]\times u[1]\times u[2] + 14\,u[0]^2\,u[3] + 42\,u[2]\times u[3] + \frac{49}{2}\,u[1]\times u[4] + 7\,u[0]\times u[5] + u[7]$

Out[136]=

0

Для каждого из них можно найти и другие высшие симметрии, как и раньше для КдФ. Немного модифицируем предыдущую процедуру.

In[137]:=

```
sym[m_, ut_] := Module[{uT, a, eq},
   uT = hom[2, 1, m + 2];
   uT = Table[a[i], {i, 1, Length[uT]}].uT /. a[1] → 1;
   eq = comm[uT, ut];
   eq = Union[Flatten[{CoefficientList[eq, vars[eq, u]]}]];
   eq = Solve[eq == 0];
   If[Length[eq] == 0, 0, uT /. eq⟦1⟧]
  ]
```

Оказывается, что симметрий 9-го порядка нет, 11-го есть. Вообще, для этих уравнений есть симметрии, порядки которых не делятся на 2 или 3.

In[138]:=

```
sym[7, uSKt[5]]
sym[9, uSKt[5]]
sym[11, uSKt[5]]
```

Out[138]=

$$\frac{28}{3} u[0]^3 u[1] + 7 u[1]^3 + 42 u[0] \times u[1] \times u[2] + 14 u[0]^2 u[3] + 21 u[2] \times u[3] + 14 u[1] \times u[4] + 7 u[0] \times u[5] + u[7]$$

Out[139]=

0

Out[140]=

$$\frac{88}{3} u[0]^5 u[1] + 330 u[0]^2 u[1]^3 + \frac{1540}{3} u[0]^3 u[1] \times u[2] + 682 u[1]^3 u[2] + 1716 u[0] \times u[1] u[2]^2 + \frac{220}{3} u[0]^4 u[3] + 1298 u[0] u[1]^2 u[3] + 990 u[0]^2 u[2] \times u[3] + 1320 u[2]^2 u[3] + 979 u[1] u[3]^2 + 616 u[0]^2 u[1] \times u[4] + 1562 u[1] \times u[2] \times u[4] + 880 u[0] \times u[3] \times u[4] + \frac{242}{3} u[0]^3 u[5] + 396 u[1]^2 u[5] + 616 u[0] \times u[2] \times u[5] + 286 u[4] \times u[5] + 286 u[0] \times u[1] \times u[6] + 220 u[3] \times u[6] + 44 u[0]^2 u[7] + 121 u[2] \times u[7] + 44 u[1] \times u[8] + 11 u[0] \times u[9] + u[11]$$

In[141]:=

```
sym[7, uKKt[5]]
sym[9, uKKt[5]]
sym[11, uKKt[5]]
```

Out[141]=

$$\frac{28}{3} u[0]^3 u[1] + \frac{35 u[1]^3}{2} + 63 u[0] \times u[1] \times u[2] + 14 u[0]^2 u[3] + 42 u[2] \times u[3] + \frac{49}{2} u[1] \times u[4] + 7 u[0] \times u[5] + u[7]$$

Out[142]=

0

Out[143]=

$$\frac{88}{3} u[0]^5 u[1] + 495 u[0]^2 u[1]^3 + \frac{1870}{3} u[0]^3 u[1] \times u[2] + \frac{3179}{2} u[1]^3 u[2] + 2871 u[0] \times u[1] u[2]^2 + \frac{220}{3} u[0]^4 u[3] + 2123 u[0] u[1]^2 u[3] + 1320 u[0]^2 u[2] \times u[3] + 2310 u[2]^2 u[3] + 1771 u[1] u[3]^2 + 781 u[0]^2 u[1] \times u[4] + \frac{5731}{2} u[1] \times u[2] \times u[4] + 1375 u[0] \times u[3] \times u[4] + \frac{242}{3} u[0]^3 u[5] + 726 u[1]^2 u[5] + 913 u[0] \times u[2] \times u[5] + 550 u[4] \times u[5] + 385 u[0] \times u[1] \times u[6] + 385 u[3] \times u[6] + 44 u[0]^2 u[7] + 187 u[2] \times u[7] + \frac{121}{2} u[1] \times u[8] + 11 u[0] \times u[9] + u[11]$$

5 Оператор рекурсии для КдФ

$R = D^2 - 4\, u - 2\, u_x\, D^{-1}$

In[144]:=

```
R[f_] := Expand[dx[f, 2] - 4 u[0] f - 2 u[1] × int[f]]

Ut[1] = uKt[1]
Ut[3] = R[%]
Ut[5] = R[%]
Ut[7] = R[%]
Ut[9] = R[%]
Ut[11] = R[%]
Ut[13] = R[%]
Ut[15] = R[%]
```

Out[145]=

```
u[1]
```

Out[146]=

```
-6 u[0] × u[1] + u[3]
```

Out[147]=

```
30 u[0]^2 u[1] - 20 u[1] × u[2] - 10 u[0] × u[3] + u[5]
```

Out[148]=

```
-140 u[0]^3 u[1] + 70 u[1]^3 + 280 u[0] × u[1] × u[2] +
 70 u[0]^2 u[3] - 70 u[2] × u[3] - 42 u[1] × u[4] - 14 u[0] × u[5] + u[7]
```

Out[149]=

```
630 u[0]^4 u[1] - 1260 u[0] u[1]^3 - 2520 u[0]^2 u[1] × u[2] + 1302 u[1] u[2]^2 -
 420 u[0]^3 u[3] + 966 u[1]^2 u[3] + 1260 u[0] × u[2] × u[3] + 756 u[0] × u[1] × u[4] -
 252 u[3] × u[4] + 126 u[0]^2 u[5] - 168 u[2] × u[5] - 72 u[1] × u[6] - 18 u[0] × u[7] + u[9]
```

Out[150]=

```
-2772 u[0]^5 u[1] + 13 860 u[0]^2 u[1]^3 + 18 480 u[0]^3 u[1] × u[2] -
 14 784 u[1]^3 u[2] - 28 644 u[0] × u[1] u[2]^2 + 2310 u[0]^4 u[3] - 21 252 u[0] u[1]^2 u[3] -
 13 860 u[0]^2 u[2] × u[3] + 9702 u[2]^2 u[3] + 7194 u[1] u[3]^2 - 8316 u[0]^2 u[1] × u[4] +
 11 484 u[1] × u[2] × u[4] + 5544 u[0] × u[3] × u[4] - 924 u[0]^3 u[5] + 2838 u[1]^2 u[5] +
 3696 u[0] × u[2] × u[5] - 924 u[4] × u[5] + 1584 u[0] × u[1] × u[6] - 660 u[3] × u[6] +
 198 u[0]^2 u[7] - 330 u[2] × u[7] - 110 u[1] × u[8] - 22 u[0] × u[9] + u[11]
```

Out[151]=

```
12 012 u[0]^6 u[1] - 120 120 u[0]^3 u[1]^3 + 30 030 u[1]^5 - 120 120 u[0]^4 u[1] × u[2] +
 384 384 u[0] u[1]^3 u[2] + 372 372 u[0]^2 u[1] u[2]^2 - 177 320 u[1] u[2]^3 - 12 012 u[0]^5 u[3] +
 276 276 u[0]^2 u[1]^2 u[3] + 120 120 u[0]^3 u[2] × u[3] - 392 964 u[1]^2 u[2] × u[3] -
 252 252 u[0] u[2]^2 u[3] - 187 044 u[0] × u[1] u[3]^2 + 26 598 u[3]^3 + 72 072 u[0]^3 u[1] × u[4] -
 77 220 u[1]^3 u[4] - 298 584 u[0] × u[1] × u[2] × u[4] - 72 072 u[0]^2 u[3] × u[4] +
 126 984 u[2] × u[3] × u[4] + 37 466 u[1] u[4]^2 + 6006 u[0]^4 u[5] - 73 788 u[0] u[1]^2 u[5] -
 48 048 u[0]^2 u[2] × u[5] + 42 042 u[2]^2 u[5] + 62 348 u[1] × u[3] × u[5] + 24 024 u[0] × u[4] × u[5] -
 20 592 u[0]^2 u[1] × u[6] + 35 464 u[1] × u[2] × u[6] + 17 160 u[0] × u[3] × u[6] - 3432 u[5] × u[6] -
 1716 u[0]^3 u[7] + 6578 u[1]^2 u[7] + 8580 u[0] × u[2] × u[7] - 2574 u[4] × u[7] + 2860 u[0] × u[1] × u[8] -
 1430 u[3] × u[8] + 286 u[0]^2 u[9] - 572 u[2] × u[9] - 156 u[1] × u[10] - 26 u[0] × u[11] + u[13]
```

Out[152]=

$$
\begin{aligned}
&-51\,480\,u[0]^{7}\,u[1]+900\,900\,u[0]^{4}\,u[1]^{3}-900\,900\,u[0]\,u[1]^{5}+720\,720\,u[0]^{5}\,u[1]\times u[2]-\\
&5\,765\,760\,u[0]^{2}\,u[1]^{3}\,u[2]-3\,723\,720\,u[0]^{3}\,u[1]\,u[2]^{2}+4\,178\,460\,u[1]^{3}\,u[2]^{2}+\\
&5\,319\,600\,u[0]\times u[1]\,u[2]^{3}+60\,060\,u[0]^{6}\,u[3]-2\,762\,760\,u[0]^{3}\,u[1]^{2}\,u[3]+1\,540\,110\,u[1]^{4}\,u[3]-\\
&900\,900\,u[0]^{4}\,u[2]\times u[3]+11\,788\,920\,u[0]\,u[1]^{2}\,u[2]\times u[3]+3\,783\,780\,u[0]^{2}\,u[2]^{2}\,u[3]-\\
&2\,279\,420\,u[2]^{3}\,u[3]+2\,805\,660\,u[0]^{2}\,u[1]\,u[3]^{2}-5\,045\,040\,u[1]\times u[2]\,u[3]^{2}-797\,940\,u[0]\,u[3]^{3}-\\
&540\,540\,u[0]^{4}\,u[1]\times u[4]+2\,316\,600\,u[0]\,u[1]^{3}\,u[4]+4\,478\,760\,u[0]^{2}\,u[1]\times u[2]\times u[4]-\\
&4\,010\,292\,u[1]\,u[2]^{2}\,u[4]+720\,720\,u[0]^{3}\,u[3]\times u[4]-2\,961\,816\,u[1]^{2}\,u[3]\times u[4]-\\
&3\,809\,520\,u[0]\times u[2]\times u[3]\times u[4]-1\,123\,980\,u[0]\times u[1]\,u[4]^{2}+578\,006\,u[3]\,u[4]^{2}-\\
&36\,036\,u[0]^{5}\,u[5]+1\,106\,820\,u[0]^{2}\,u[1]^{2}\,u[5]+480\,480\,u[0]^{3}\,u[2]\times u[5]-1\,959\,672\,u[1]^{2}\,u[2]\times u[5]-\\
&1\,261\,260\,u[0]\,u[2]^{2}\,u[5]-1\,870\,440\,u[0]\times u[1]\times u[3]\times u[5]+480\,194\,u[3]^{2}\,u[5]-\\
&360\,360\,u[0]^{2}\,u[4]\times u[5]+763\,620\,u[2]\times u[4]\times u[5]+187\,174\,u[1]\,u[5]^{2}+205\,920\,u[0]^{3}\,u[1]\times u[6]-\\
&274\,560\,u[1]^{3}\,u[6]-1\,063\,920\,u[0]\times u[1]\times u[2]\times u[6]-257\,400\,u[0]^{2}\,u[3]\times u[6]+\\
&543\,400\,u[2]\times u[3]\times u[6]+320\,632\,u[1]\times u[4]\times u[6]+102\,960\,u[0]\times u[5]\times u[6]+\\
&12\,870\,u[0]^{4}\,u[7]-197\,340\,u[0]\,u[1]^{2}\,u[7]-128\,700\,u[0]^{2}\,u[2]\times u[7]+134\,706\,u[2]^{2}\,u[7]+\\
&199\,888\,u[1]\times u[3]\times u[7]+77\,220\,u[0]\times u[4]\times u[7]-12\,870\,u[6]\times u[7]-42\,900\,u[0]^{2}\,u[1]\times u[8]+\\
&88\,400\,u[1]\times u[2]\times u[8]+42\,900\,u[0]\times u[3]\times u[8]-10\,010\,u[5]\times u[8]-2860\,u[0]^{3}\,u[9]+\\
&13\,130\,u[1]^{2}\,u[9]+17\,160\,u[0]\times u[2]\times u[9]-6006\,u[4]\times u[9]+4680\,u[0]\times u[1]\times u[10]-\\
&2730\,u[3]\times u[10]+390\,u[0]^{2}\,u[11]-910\,u[2]\times u[11]-210\,u[1]\times u[12]-30\,u[0]\times u[13]+u[15]
\end{aligned}
$$

Результаты совпадают с предыдущими:

In[153]:=

```
Table[uKt[j] - Ut[j], {j, 1, 11, 2}]
```

Out[153]=

```
{0, 0, 0, 0, 0, 0}
```